
Abstract
Advances in quantum computing have shown a serious challenge for widely used current cryptographic
techniques because a sufficient large-scale quantum computer can efficiently solve hard mathematical
problems on which the current public-key cryptography is relying. That is the reason why recently
many researchers and industrial companies have spent lots of effort on constructing post-quantum cryp-
tosystems, which are resistant to quantum attackers. Large numbers of post-quantum key encapsulation
mechanisms (KEMs) have been proposed to provide secure key establishment - one of the most important
building blocks in asymmetric cryptography. This paper presents a formal security analysis of three
lattice-based KEMs: Kyber, Saber, and SK-MLWR. We first formally specify each of them in Maude, a
rewriting logic-based specification and programming language equipped with many functionalities, such
as a reachability analyzer (or the search command) that can be used as an invariant model checker, and
then conduct invariant model checking with the Maude search command, finding an attack.

1. Introduction

In recent years, advanced research in the field of quantum computing and quantum information
theory has brought a credible threat to cryptosystems currently in use. The most popular
asymmetric (or public-key) primitives used today will become insecure against sufficiently

strong quantum computers because they can be efficiently broken by Shor’s algorithm [1]. The
security of these primitives relies on one of the following three hard mathematical problems:
the integer factorization problem, the discrete logarithm problem, and the elliptic-curve discrete
logarithm problem. All of these problems are hard under conventional computers, but they
can be easily solved on a sufficiently powerful quantum computer running Shor’s algorithm.
On the other hand, symmetric primitives are considered secure against quantum attackers.
Although Grover’s algorithm [2], one of the most well-known quantum algorithms, can reduce
the complexity to break symmetric primitives, doubling the key size can efficiently avoid these
attacks. For example, we can say that AES-256 would be as hard to break by a quantum computer
as AES-128 is by a classical computer.

As a response to the quantum attack threat, there is extensive research to find new schemes
which are secure even in the presence of quantum adversaries. In the past few years, many
post-quantum asymmetric primitives have been proposed as replacements for those traditional
ones currently in use. The National Institute of Standards and Technology of USA (NIST) also
started the Post-Quantum Cryptography Project in 2017, calling for proposals of post-quantum
cryptographic protocols that are secure against both conventional and quantum computers1.
There were 82 submissions to this standardization project, implying the importance of this
problem. Among these submissions, a large number of proposals are for post-quantum key
encapsulation mechanisms (KEMs), which aim to securely establish a symmetric key between
two parties. This is understandable because key exchange algorithms are considered the most
important building block of asymmetric cryptography.

Security analysis of cryptographic primitives and/or protocols can be fundamentally divided
into two approaches: computational security and symbolic security. Proof in the computational
model requires a definition of secure cryptographic construction (primitive, protocol), and some
assumptions about the computationally infeasible problem. The proof can be regarded as a
mathematical reduction to the situation where the only chance to violate the security of such a
construction is to solve the infeasible problem. The authors of the three KEMs considered in this
paper have already presented their security proofs in the computational model. However, such
proofs are often not easy to understand for non-experts in cryptography. On the other hand,
symbolic analysis is easier to understand, computer-verified and suitable for automation. Our
approach presented in this paper belongs to the latter. Note that our approach can be applied to
not only the three KEMs but also other KEMs and other kinds of primitives as well.

We formally specify and model check three KEMs: Kyber [3] (precisely CRYSTALS-Kyber),
Saber [4], and SK-MLWR [5]. Because of space limitation, we choose Kyber as the only KEM to
illustrate in this paper. The specifications of the other KEMs can be found at https://github.com/
duongtd23/kems-mc. Kyber is a KEM whose security is based on the hardness of solving the
learning-with-errors (LWE) problem, while the security of Saber and SK-MLWR relies on the
hardness of the Module Learning With Rounding (MLWR) problem. All of them belong to the
lattice-based cryptography. We use Maude [6], a programming/specification language based on
rewriting logic, to formally specify the Dolev-Yao generic intruder [7] as well as these KEMs.
By employing the Maude search command, a Man-In-The-Middle (MITM) attack is found for
each KEM. Although this kind of attack is not a novel attack for KEMs, the formal specifications

1https://csrc.nist.gov/projects/post-quantum-cryptography

in Maude and the model checking experiments are worth reporting. Our ultimate goal is to
come up with a new security analysis/verification technique for post-quantum cryptographic
protocols, which use post-quantum cryptographic primitives, such as the three KEMs reported
in this paper. Formally specifying such primitives is necessary for analyzing the security later
on. What is described in the paper is our initial step toward the goal.

Related work. In 2012, Blanchet [8] has surveyed various approaches to security protocol
verification in both symbolic model and computational model. In the symbolic model, there
is a large number of tools existing for verifying cryptosystems, such as ProVerif [9], Maude-
NPA [10], Tamarin [11], and Scyther [12]. The symbolic protocol verifier ProVerif, which was
developed by Blanchet, can automatically prove security properties of cryptographic protocol
specifications. ProVerif is based on an abstract representation of the protocol by a set of Horn
clauses, and it determines whether the desired security properties hold by resolution on these
clauses. The practicability of ProVerif has been demonstrated through case studies, such as
[13, 14]. ProVerif can handle an unbounded number of sessions (executions) of protocols, but
termination is not guaranteed in general because the resolution algorithm may not terminate.
To mitigate this challenge, Escobar et al. [15] proposed some techniques to reduce the size
of the search space in Maude-NPA, such as generating formal grammars representing terms
(states information) unreachable from initial states and using super lazy intruder to delay the
generation of substitution instances as much as possible. Even though, the termination of the
tool is not always guaranteed. Among many case studies that demonstrated the capabilities of
Maude-NPA, [16] presented one case study with Diffie-Hellman key agreement protocol.

Scyther [12] is another tool for symbolic security verification of cryptographic protocols.
Like ProVerif, Scyther also supports an unbounded number of sessions, but it supports only a
fixed set of cryptographic primitives (symmetric and asymmetric encryption and signatures)
and does not allow for user-specified equational theories. Its successor, namely Tamarin [11]
prover, does support equational theories. Moreover, Tamarin provides two ways of constructing
proofs: fully automated mode and interactive mode. The tool may not terminate in the fully
automated mode. In the interactive mode, the tool allows users to provide lemmas that must
be proved. Several case studies on security analysis of cryptographic primitives and protocols
with Tamarin can be found in [17, 18].

Yadav et al. [19] explored NTRU key exchange, a lattice-based public key exchange protocol,
and found that it is exposed to an MITM attack. The attack was found in the same manner as
what we present in this paper. However, they used neither any tool nor formal specification
language as we do.

In the computational security approach, game-based model is known as a standard model for
proving security. Security for cryptographic primitives or protocols is defined as an attack game
played between an adversary and some benign entity, which is called the challenger. The main
idea of the game-based security model is simulation of interaction among these two parties.
Eventually, the security proof typically leads to a proof that any supposed adversary can get
an advantage over the challenger if and only if he/she is able to solve some computationally
infeasible problem (e.g., discrete logarithm, integer factorization). When a proof becomes too
complicated, the proof normally employs the sequence of games technique [20]. CryptoVerif [21]
is a tool for mechanizing such proof. It can generate proofs by sequences of games automatically



or with little user interaction. Alwen et al. [22] have employed CryptoVerif to analyze the
security of the Hybrid Public Key Encryption (HPKE), which is a candidate for a new public
key encryption standard.

Roadmap. The remaining of this paper is organized as follows: Section 2 gives some prelimi-
naries, such as KEM and state machine. Section 3 describes Kyber KEM, briefly explains the
underlying learning with error problem. The specification of Kyber in Maude is presented in
Section 4. The model checking result and the attack found are presented in Section 5. Finally,
Section 6 summarizes the paper.

2. Preliminaries

2.1. Key encapsulation mechanism

A key encapsulation mechanism is a tuple of algorithms (KeyGen, Encaps, Decaps) along with a
finite keyspace 𝒦:

• KeyGen()→ (𝑝𝑘, 𝑠𝑘): A probabilistic key generation algorithm that outputs a public key
𝑝𝑘 and a secret key 𝑠𝑘.

• Encaps(𝑝𝑘)→ (𝑐, 𝑘): A probabilistic encapsulation algorithm that takes as input a public
key 𝑝𝑘, and outputs a ciphertext (or encapsulation) 𝑐 and a key 𝑘 ∈ 𝒦.

• Decaps(𝑐, 𝑠𝑘)→ 𝑘: A (usually deterministic) decapsulation algorithm that takes as input
a ciphertext 𝑐 and a secret key 𝑠𝑘, and outputs a key 𝑘 ∈ 𝒦.

2.2. State machine and Maude

A state machineℳ ≜ ⟨𝒮, ℐ, 𝒯 ⟩ consists of a set 𝒮 of states, a set ℐ ⊆ 𝒮 of initial states and
a binary relation 𝒯 ⊆ 𝒮 × 𝒮 over states. The set ℛ of reachable states with respect toℳ is
inductively defined as follows: (1) for each 𝑠 ∈ ℐ , 𝑠 ∈ ℛ and (2) for each (𝑠, 𝑠′) ∈ 𝑇 , if 𝑠 ∈ ℛ,
then 𝑠′ ∈ ℛ. A state predicate 𝑝 is an invariant property with respect toℳ if and only if 𝑝(𝑠)
holds for all 𝑠 ∈ ℛ.

In this paper, to express a state of 𝒮 , we use a braced associative-commutative collection
of name-value pairs. Associative-commutative collections are called soups, and name-value
pairs are called observable components. That is, a state is expressed as a soup of observable
components. The juxtaposition operator is used as the constructor of soups. Let 𝑜𝑐1, 𝑜𝑐2, 𝑜𝑐3 be
observable components, and then 𝑜𝑐1 𝑜𝑐2 𝑜𝑐3 is the soup of those three observable components.
A state is of the form {𝑜𝑐1 𝑜𝑐2 𝑜𝑐3}. There are multiple possible ways to specify state transitions.
In this paper, we use Maude [6], a programming and specification language based on rewriting
logic, to specify them as rewrite rules. Maude makes it possible to specify complex systems
flexibly and it is also equipped with model checking facilities (a reachability analyzer and an
LTL model checker). A rewrite rule starts with the keyword rl, followed by a label enclosed
with square brackets and a colon, two patterns connected with =>, and ends with a full stop.
A conditional one starts with the keyword crl and has a condition following the keyword if

before a full stop. The following is a form of a conditional rewrite rule:



crl [𝑙𝑏] : 𝑙 => 𝑟 if . . . /\ 𝑐𝑖 /\ . . . .

where 𝑙𝑏 is a label and 𝑐𝑖 is part of the condition, which may be an equation 𝑙𝑐𝑖 = 𝑟𝑐𝑖. If the
condition . . . /\ 𝑐𝑖 /\ . . . holds under some substitution 𝜎, 𝜎(𝑙) can be replaced with 𝜎(𝑟).

Maude provides the search command that can find a state reachable from 𝑡 such that the
state matches 𝑝 and satisfies the condition(s) 𝑐:

search [n,m] in MOD : 𝑡 =>* 𝑝 such that 𝑐 .

where MOD is the name of the module specifying the state machine, and n and m are optional
arguments denoting a bound on the number of desired solutions and the maximum depth of
the search. n typically is 1 and 𝑡 typically represents an initial state of the state machine.

3. Kyber Key Encapsulation Mechanism

3.1. Notations

Let ℬ denote the set {0, . . . , 255}, i.e., the set of 8-bit unsigned integers (bytes). Consequently,
ℬ𝑘 denotes the set of byte arrays of length 𝑘 and ℬ* denotes the set of byte arrays of arbitrary
length. For two byte arrays 𝑎 and 𝑏, (𝑎||𝑏) denotes the concatenation of 𝑎 and 𝑏.

The ring of integers modulo 𝑞 is denoted by Z𝑞 . We denote by 𝑅 the polynomial ring
Z[𝑋]/(𝑋𝑛 + 1) and by 𝑅𝑞 the quotient polynomial ring Z𝑞[𝑋]/(𝑋𝑛 + 1). Thus polynomials
in 𝑅𝑞 are of 𝑛 coefficients where each coefficient is in [0, 𝑞). In Kyber, the values of 𝑛 and 𝑞
are always fixed to 𝑛 = 256 and 𝑞 = 7681 in all levels of security. Regular font letters denote
elements in 𝑅 or 𝑅𝑞 (which includes elements in Z and Z𝑞) and bold lower-case letters represent
vectors with coefficients in 𝑅 or 𝑅𝑞 . By default, all vectors are column vectors. Bold upper-case
letters are matrices. For a vector v (or matrix A), we denote by v𝑇 (or A𝑇 ) its transpose. For a
vector v we write v[𝑖] to denote its 𝑖-th entry (with indexing starting at zero); for a matrix A
we write A[𝑖][𝑗] to denote the entry in row 𝑖 and column 𝑗 (again, with indexing starting at
zero).

Let H and G be two hash functions, where H : ℬ* → ℬ32 and G : ℬ* → ℬ32×ℬ32. Let KDF
denote the key derivation function, where KDF : ℬ* → ℬ32. Let 𝑥 ∈ Q, where Q denotes the
rational numbers set, then ⌊𝑥⌉ denotes rounding of 𝑥 to the closest integer. Let 𝑥 ∈ Z𝑞 and
𝑑 < ⌊log2 𝑞⌉, functions Compress and Decompress are defined by the following equations:

Compress𝑞(𝑥, 𝑑) = ⌊(2𝑑/𝑞) · 𝑥⌉ mod 2𝑑, (1)

Decompress𝑞(𝑥, 𝑑) = ⌊(𝑞/2𝑑) · 𝑥⌉ (2)

Let 𝑥′ = Decompress𝑞(Compress𝑞(𝑥, 𝑑), 𝑑), then we have:

|𝑥′ − 𝑥 mod 𝑞| ≤ ⌊ 𝑞

2𝑑+1
⌉ (3)

When Compress𝑞 and Decompress𝑞 are used with 𝑥 ∈ 𝑅𝑞 or x ∈ 𝑅𝑘
𝑞 , they are applied to each

coefficient individually.



KEM.KeyGen()
𝑧 ← ℬ32
(𝑝𝑘, 𝑠𝑘′) = CPAPKE.KeyGen()
𝑠𝑘 = (𝑠𝑘′||𝑝𝑘||H(𝑝𝑘)||𝑧)
return (𝑝𝑘, 𝑠𝑘)

KEM.Dec(𝑐, 𝑠𝑘)
(s||𝑝𝑘||H(𝑝𝑘)||𝑧) = 𝑠𝑘
𝑚′ = CPAPKE.Dec(𝑐, s)
(𝐾̄

′
, 𝑟′) = G(𝑚′||H(𝑝𝑘))

𝑐′ = CPAPKE.Enc(𝑝𝑘,𝑚′, 𝑟′)

if 𝑐 = 𝑐′ then return 𝐾 = KDF(𝐾̄
′
,H(𝑐))

else return 𝐾 = KDF(𝑧,H(𝑐))

KEM.Enc(𝑝𝑘)
pk−−−−→ 𝑚0 ← ℬ32

𝑚 = H(𝑚0)
(𝐾̄, 𝑟) = G(𝑚||H(𝑝𝑘))
𝑐 = CPAPKE.Enc(𝑝𝑘,𝑚, 𝑟)
𝐾 = KDF(𝐾̄,H(𝑐))

c←−−−− return (𝑐,𝐾)

Figure 1: Kyber.KEM

3.2. Kyber

Fig. 1 describes the three algorithms (KeyGen, Encaps, Decaps) of Kyber KEM. It employs the
three algorithms (KeyGen, Enc, Dec) of Kyber.CPAPKE, which are shown in Fig. 2. Let us suppose
that there are two parties Alice and Bob. The interactions depicted in Fig. 1 are as follows. Alice
performs the KEM.KeyGen step, starting by choosing a random seed 𝑑, and hashing it to get the
pair (𝜌, 𝜎). From 𝜎, she generates vector s serving as the secret key 𝑠𝑘 and vector e acting as an
error component. t is then computed from s, e, and matrix A, which is generated from 𝜌. The
pair of t and 𝜌 serves as the public key 𝑝𝑘, which is sent to Bob. Upon receiving 𝑝𝑘, Bob executes
the KEM.Enc step (i.e., Encaps step) in Fig. 1. He randomly chooses an 𝑚0, hashes it, and passes
the outputs to the CPAPKE.Enc procedure (depicted in Fig. 2). The obtained ciphertext 𝑐, which
is a pair of 𝑐1 and 𝑐2, is sent back to Alice. Upon receiving 𝑐, Alice performs the KEM.Dec step
(i.e., Decaps step). She computes 𝑐′ by employing the CPAPKE.Dec and CPAPKE.Enc procedures.
With a very high probability 𝑐′ is equal to 𝑐, implying that 𝑚′ on Alice’s side is equal to 𝑚 on
Bob’s side with an overwhelming probability. After that, they can derive the same key 𝐾 . Note
that all multiplications and additions in the two figures are computed over Z𝑞[𝑋]/(𝑋𝑛 + 1).

Note that in [3], the definition of Kyber employs the functions Encode and Decode. Function
Encode serializes a polynomial or a vector of polynomials to a byte array, and function Decode
is the inverse of Encode. Furthermore, to perform multiplications in 𝑅𝑞 efficiently, the vectors
and matrices are converted to NTT domain and vice versa, where NTT stands for number-
theoretic transform. However, implementation or performance is out of the scope of the present
paper; thus, for simplicity and ease of understanding, we omit those concepts. Consequently,
the notation, e.g., 𝑝𝑘 = (t||𝜌) is a misuse of notations because t ∈ 𝑅𝑘

𝑞 and then 𝑡 is not a byte
array. This notation is understood as 𝑝𝑘 is made of t and 𝜌.

The procedure to generate matrix A, which is denoted by generate(𝜌) in Fig. 2, taking as
input a random seed 𝜌, is deterministic. Informally, if Alice and Bob share the same random
seed 𝜌, then they can agreeingly derive the same matrix A, whose coefficients of each entry are
close to a uniformly random distribution. In contrast, the procedure to sample noise (or error)



CPAPKE.KeyGen()
𝑑← ℬ32
(𝜌, 𝜎) = G(𝑑)
𝑅𝑘×𝑘

𝑞 ∋ A = generate(𝜌)

𝑅𝑘
𝑞 ∋ s, e← sampleCBD(𝜎)

t = As+ e
𝑝𝑘 = (t||𝜌)
𝑠𝑘 = s
return (𝑝𝑘, 𝑠𝑘)

CPAPKE.Dec(𝑐, 𝑠𝑘)
(𝑐1||𝑐2) = 𝑐
u′ = Decompress𝑞(𝑐1, 𝑑𝑢)
𝑣′ = Decompress𝑞(𝑐2, 𝑑𝑣)

𝑚′ = Compress𝑞(𝑣
′ − s𝑇u′, 1)

return 𝑚′

CPAPKE.Enc(𝑝𝑘,𝑚, 𝑟)
(t||𝜌) = 𝑝𝑘
𝑅𝑘×𝑘

𝑞 ∋ A = generate(𝜌)

𝑅𝑘
𝑞 ∋ r, e1 ← sampleCBD(𝑟)

𝑅𝑞 ∋ 𝑒2 ← sampleCBD(𝑟)
u = A𝑇 r+ e1
𝑣 = t𝑇 r+ 𝑒2 +Decompress𝑞(𝑚, 1)
𝑐1 = Compress𝑞(u, 𝑑𝑢)
𝑐2 = Compress𝑞(v, 𝑑𝑣)
return 𝑐 = (𝑐1||𝑐2)

Figure 2: Kyber.CPAPKE

components (e.g., e, e1, and 𝑒2), namely sampleCBD, is probabilistic. It takes as input a random
seed (e.g., 𝜌 and 𝑟) and returns a polynomial whose coefficients are close to a centered binomial
distribution (to sample a vector, e.g., s and e, the procedure is called 𝑘 times). Informally,
coefficients of an output of sampleCBD are mostly close to 0, and their absolute value is never
greater than a specific small number (which is 5, or 4, or 3, depending on the level of security).

4. Formal specification of Kyber

4.1. Formalization of polynomials, vectors, and matrices

We first introduce sort Poly that represents polynomials as follows:

sort Poly . subsort Int < Poly .
op _p+_ : Poly Poly -> Poly [ctor assoc comm prec 33] .
op _p*_ : Poly Poly -> Poly [ctor assoc comm prec 31] .
op _p-_ : Poly Poly -> Poly [prec 33] .
op neg_ : Poly -> Poly [ctor] .

where Int is the sort of integers. The notation subsort Int < Poly indicates that any inte-
ger is also a polynomial. p+, p*, and p- denote the addition, multiplication, and subtraction,
respectively, between two polynomials. neg denotes the negation of a polynomial. assoc

comm indicates that _p+_ and _p*_ are declared to be associative and commutative. prec 33

attached with _p+_ and _p-_ indicates that these operators have the same precedence 33, which
is lower precedence than that of _p*_ (i.e., 31). Note that, we only consider polynomials in
Z𝑞[𝑋]/(𝑋𝑛 + 1) (or 𝑅𝑞), where 𝑛 = 256 and 𝑞 = 7681. Let P1, P2, and P3 be variables of Poly.
We declare some properties of the operators as follows:

eq P1 p+ 0 = P1 . eq P1 p* 0 = 0 . eq P1 p* 1 = P1 .
eq P1 p* (P2 p+ P3) = (P1 p* P2) p+ (P1 p* P3) .



eq P1 p- P2 = P1 p+ neg(P2) . eq P1 p+ neg(P1) = 0 .
eq neg(neg(P1)) = P1 . eq neg(P1 p+ P2) = neg(P1) p+ neg(P2) .

In a similar way, we introduce sorts Vector and Matrix representing polynomial vectors and
matrices, respectively; operators v+, dot, and m* representing the addition & inner product of
two polynomial vectors, and multiplication of a polynomial matrix and a vector, respectively. Let
V1, V2, and V3 be variables of Vector. The declarations of the three operators and the distributive
property of vectors are specified as follows:

op _v+_ : Vector Vector -> Vector [assoc comm prec 33] .
op _dot_ : Vector Vector -> Poly [prec 31] .
op _m*_ : Matrix Vector -> Vector [prec 31] .
eq (V1 v+ V2) dot V3 = (V1 dot V3) p+ (V2 dot V3) .
eq V3 dot (V1 v+ V2) = (V3 dot V1) p+ (V3 dot V2) .

4.2. Formalization of honest parties

Two constructors for the two kinds of messages used in Kyber are as follows:

op msg1 : Prin Prin Prin Vector Poly MsgState -> Msg [ctor] .
op msg2 : Prin Prin Prin Vector Poly MsgState -> Msg [ctor] .

where Prin is the sort representing principals, and Msg is the sort denoting messages. MsgState
is the sort representing message states, receiving one of the following three values: sent -
the message was sent, replied - the message was sent and the receiver replied with another
message, and intercepted - the message was intercepted by the intruder. The first, second, and
third arguments of each operator are the actual creator, the seeming sender, and the receiver
of the corresponding message. The first and last arguments are meta-information that is only
available to the outside observer, while the remaining arguments can be seen by every principal.
The fourth and fifth arguments of msg1 carry the vector t and the random seed 𝜌, respectively,
of the public key 𝑝𝑘 (𝑝𝑘 is (t||𝜌) as explained in Section 3.2). Similarly, the fourth and fifth
arguments of msg2 carry 𝑐1 and 𝑐2, respectively, of CPAPKE.

We model the network as a multiset of messages, in which the intruder can use as his/her
storage. Consequently, the empty network (i.e., the empty multiset) means that no messages
have been sent. The intruder can fully control the network, that is he/she can intercept any
message, glean information from it, and fake a new message to any honest party. To formally
specify Kyber in Maude, we use the following observable components:

• (nw : 𝑚𝑠𝑔𝑠) - 𝑚𝑠𝑔𝑠 is the soup of messages in the network;

• (keys[p] : 𝑘𝑒𝑦𝑠) - 𝑘𝑒𝑦𝑠 is a soup of the computed shared keys of principal p. Each entry
of 𝑘𝑒𝑦𝑠 is in form of key(K,q), where K is the shared key and q is the principal whom p

believes that he/she has communicated with;

• (prins : 𝑝𝑠) - 𝑝𝑠 is the collection of all principals participating in the protocol;

• (d[p] : 𝑑0) - 𝑑0 is the random seed 𝑑 (used in Fig. 2) of principal p;

• (m[p] : 𝑚0) - 𝑚0 is the random seed 𝑚0 (shown in Fig. 1)) of principal p;



• (rd-d : 𝑟𝑑𝑑𝑠) - 𝑟𝑑𝑑𝑠 is a list of available values as the random seed 𝑑 (we use list, but not
set, to reduce the state space for searching). Each time when a principal makes a query
for a random value of 𝑑, the top value in 𝑟𝑑𝑑𝑠 is removed and returned to the principal;

• (rd-m : 𝑟𝑑𝑚𝑠) - 𝑟𝑑𝑚𝑠 is a list of the available values as random seed 𝑚0;

• (glean-keys : 𝑔𝑘𝑒𝑦𝑠) - 𝑔𝑘𝑒𝑦𝑠 is the soup of shared keys gleaned by the intruder;

• (ds : 𝑑𝑠) - 𝑑𝑠 is the collection of the random seeds 𝑑 used by the intruder. Note that every
entry in 𝑑𝑠 is different from any random value used by honest parties;

• (ms : 𝑚𝑠) - 𝑚𝑠 is the collection of random seeds m used by the intruder. Similarly to ds,
every entry in 𝑚𝑠 is different from any random value used by honest parties.

Each state in𝒮Kyber is expressed as {𝑜𝑏𝑠}, where 𝑜𝑏𝑠 is a soup of those observable components.
We suppose that there are two honest principals alice and bob together with a malicious one,
namely eve, participating in Kyber KEM. The initial state init of ℐKyber is defined as follows:

{(nw: empty) (prins: (alice bob eve)) (rd-d: (d1 , d2)) (rd-m: (m1 , m2))
(keys[alice]: empty) (keys[bob]: empty) (glean-keys: empty) (d[alice]: 0)
(d[bob]: 0) (m[alice]: 0) (m[bob]: 0) (ds: empty) (ms: empty)} .

With the honest parties, we specify three transitions: keygen, encaps, and decaps, which
correspond to the three steps of the mechanism. Let OCs be a variable of observable component
soups, A, B, and C be variables of principals (possibly intruder), and PS be a variable of principal
collections. Let D, M, M2, M’, Rho, RhoA, V, V’, CV, and P1 be variables of polynomials, and PoL be a
variable of polynomial lists. Let G and H denote the hash functions G and H, respectively. Let MS
be a variable of networks (i.e., message soups). The rewrite rule keygen is defined as follows:

crl [keygen] : {(rd-d: (D, PoL)) (d[A]: P1) (prins: (A B PS)) (nw: MS) OCs}
=> {(rd-d: PoL) (d[A]: D) (prins: (A B PS))

(nw: (MS msg1(A,A,B, sample-A(1st(RhoSig)) m* sample-s(2nd(RhoSig))
v+ sample-e(2nd(RhoSig)), 1st(RhoSig), sent))) OCs}

if RhoSig := G(D) .

where RhoSig is a variable denoting a pair of polynomials, 1st and 2nd are its projection
operators. sample-A, sample-e, and sample-s represent the sampling procedures, outputting
the matrix A, the vectors e, and s, respectively. The rewrite rule says that when there exists a
polynomial D in rd-d, A picks it as a random seed 𝑑, builds a message msg1 exactly following the
KeyGen() step of the mechanism, and sends it to B. d[A] is set to D, and D is removed from rd-d.

The rewrite rule encap is defined as follows:

crl [encap] : {(rd-m: (M, PoL)) (m[B]: P1) (keys[B]: KS)
(nw: (msg1(C,A,B,T,Rho,sent) MS)) OCs}

=> {(rd-m: PoL) (m[B]: M) (keys[B]: (KS key(KDF(1st(Kr), H’(CU,CV)), A)))
(nw: (msg1(C,A,B,T,Rho,replied) msg2(B,B,A, CU, CV, sent) MS)) OCs}

if M’ := H(M) /\ Kr := G(pair(M’, H’(T, Rho))) /\
CU := enc-u(T, Rho, M’, 2nd(Kr)) /\ CV := enc-v(T, Rho, M’, 2nd(Kr)) .



where T and CU are variables of polynomial vectors, Kr is a variable denoting a pair of polynomials,
and KS is a variable representing a soup of shared keys. Let Rseed be a variable of polynomials.
Let sample-r, sample-e1, and sample-e2 represent the procedures sampling r, e1, and 𝑒2,
respectively. Following the CPAPKE.Enc(𝑝𝑘,𝑚, 𝑟) in Fig. 2, enc-u and enc-v are defined as
follows:

eq enc-u(T,Rho,M,Rseed) =
compr(tp(sample-A(Rho)) m* sample-r(Rseed) v+ sample-e1(Rseed),du) .

eq enc-v(T,Rho,M,Rseed) =
compr(tpV(T) dot sample-r(Rseed) p+ sample-e2(Rseed) p+ decompr(M,1),dv) .

where tp(A) denotes the transpose matrix of A and tpV(T) denotes the transpose vector of T.
du and dv are constants of natural numbers, denoting 𝑑𝑢 and 𝑑𝑣, respectively. decompr(M,1)
and compr(M,1) denote Decompress𝑞(M, 1) and Compress𝑞(M, 1), respectively. enc-u(T,Rho,M
,Rseed) and enc-v(T,Rho,M,Rseed) compute 𝑐1 and 𝑐2, respectively, in Fig. 2 given as inputs
T||Rho, M, Rseed. The rewrite rule encaps says that when there exists a message msg1 sent from
A to B in the network, B builds a message msg2 exactly following the Encaps() step of Kyber,
sends it back to A. B also computes the shared key with A, and the state of the message msg1 is
updated to replied.

The rewrite rule decaps is defined as follows:

crl [decaps] : {(d[A]: D) (keys[A]: KS)
(nw: (msg1(A,A,B,T,Rho,MsgStat) msg2(C,B,A, CU, CV, sent) MS)) OCs}

=> {(d[A]: D) (keys[A]: (KS key(KDF(1st(Kr2), H’(CU,CV)), B)))
(nw: (msg1(A,A,B,T,Rho,MsgStat) msg2(C,B,A, CU, CV, replied) MS)) OCs}

if RhoSig := G(D) /\ Rho == 1st(RhoSig) /\
T == sample-A(Rho) m* sample-s(2nd(RhoSig)) v+ sample-e(2nd(RhoSig)) /\
U’ := decompr(CU, du) /\ V’ := decompr(CV, dv) /\
M2 := compr(V’ p- tpV(sample-s(2nd(RhoSig))) dot U’, 1) /\
Kr2 := G(pair(M2, H’(T, Rho))) /\
enc-u(T,Rho,M2,2nd(Kr2)) == CU /\ enc-v(T,Rho,M2,2nd(Kr2)) == CV .

where MsgStat is a variable representing an arbitrary message state. The rewrite rule says that
when A has sent a message msg1 to B and there exists a message msg2 replied from B to A in
the network, A follows the Decaps() step of Kyber, computes the shared key with B. We only
consider the overwhelming case, i.e., Alice successfully recovers 𝑚. We assume that the error
tolerance gaps made by error components always be silent, making 𝑚′ equals to 𝑚. This is
done by the following equation:

ceq compr(E0 p+ decompr(M,1),1) = M if isSmall?(E0) .

where E0 is a variable of sort Poly, and isSmall?(E0) is a predicate, returning true if all co-
efficients of E0 are small in comparison with 𝑞. Sampling procedures for s, e, r, e1, and 𝑒2
return vectors or polynomials whose coefficients are small. These properties are specified by
the following equations:

eq isSmall?(sample-s(P)) = true . eq isSmall?(sample-e(P)) = true .
eq isSmall?(sample-e1(P)) = true . eq isSmall?(sample-e2(P)) = true .
eq isSmall?(sample-r(P)) = true .



Using Eq. 3, we rewrite Decompress𝑞(Compress𝑞(𝑣, 𝑑𝑣), 𝑑𝑣) and Decompress𝑞(
Compress𝑞(u, 𝑑𝑢), 𝑑𝑢) by 𝑣+ 𝜖1 and u+𝜖2, respectively, where all coefficients of 𝜖1 and 𝜖2 are
small in comparison with those of 𝑣 and u. In the specification, we specify 𝜖1 as epsilon1(𝑣), 𝜖2
as epsilon2(u), and both epsilon1(𝑣) & epsilon2(u) are “small”. This is done by the following
equations:

eq decompr(compr(V,dv),dv) = V p+ epsilon1(V) .
eq decompr(compr(U,du),du) = U v+ epsilon2(U) .
eq isSmall?(epsilon1(V)) = true . eq isSmall?(epsilon2(U)) = true .

4.3. Formalization of intruders

We suppose that there is one intruder, namely eve, participating in the mechanism. When there
exists a message msg1 sent from A to B in the network, the intruder can intercept that message,
fake a new message, and send it to the receiver. This behavior is specified by the following
rewrite rule:

crl [keygen-eve] : {(ds: (D PoC1)) (nw: (msg1(A,A,B,TA,RhoA,sent) MS)) OCs}
=> {(ds: (D PoC1)) (nw: (msg1(A,A,B,TA,RhoA,intercepted)

msg1(eve,A,B,sample-A(1st(RhoSig)) m* sample-s(2nd(RhoSig)) v+
sample-e(2nd(RhoSig)),1st(RhoSig),sent) MS)) OCs}

if RhoSig := G(D) .

where PoC1 and PoC3 are variables representing arbitrary soups of polynomials. The intercepted
message must have state sent at the beginning, which means that the message has not reached
the receiver. eve then constructs a new faking message from an available value D for the random
seed 𝑑. This kind of random value cannot be gleaned from the network, but eve can only
construct it by randomly choosing a new value as the rewrite rule build-ds as follows:

rl [build-ds] : {(rd-d: (D, PoL)) (ds: PoC1) OCs}
=> {(rd-d: PoL) (ds: (PoC1 D)) OCs} .

Similarly, the only way in which eve can construct values for the random seed 𝑚 is by randomly
choosing a new value. This is specified by the following rewrite rule build-ms:

rl [build-ms] : {(rd-m: (M, PoL)) (ms: PoC3) OCs}
=> {(rd-m: PoL) (ms: (PoC3 M)) OCs} .

Two more rewrite rules are introduced as follows:

crl [encaps-eve] : {(ms: (M PoC3)) (glean-keys: KS)
(nw: (msg1(A,A,B,TA,RhoA,intercepted) MS)) OCs}

=> {(ms: (M PoC3)) (glean-keys: (key(KDF(1st(Kr),H’(CU,CV)),A) KS))
(nw: (msg1(A,A,B,TA,RhoA,intercepted) msg2(eve,B,A,CU,CV,sent) MS)) OCs}

if M’ := H(M) /\ Kr := G(pair(M’,H’(TA,RhoA))) /\
CU := enc-u(TA,RhoA,M’,2nd(Kr)) /\ CV := enc-v(TA,RhoA,M’,2nd(Kr)) /\
msg2(eve,B,A,CU,CV,sent) \in MS = false /\
msg2(eve,B,A,CU,CV,replied) \in MS = false .

crl [decaps-eve] : {(ds: (D PoC1)) (glean-keys: KS)
(nw: (msg1(eve,A,B,T,Rho,replied) msg2(B,B,A,CUB,CVB,sent) MS)) OCs}



=> {(ds: (D PoC1)) (glean-keys: (key(KDF(1st(Kr2), H’(CUB, CVB)), B) KS))
(nw: (msg1(eve,A,B,T,Rho,replied) msg2(B,B,A,CUB,CVB,intercepted) MS)) OCs}

if RhoSig := G(D) /\ Rho == 1st(RhoSig) /\
T == sample-A(Rho) m* sample-s(2nd(RhoSig)) v+ sample-e(2nd(RhoSig)) /\
UB’ := decompr(CUB, du) /\ VB’ := decompr(CVB, dv) /\
M2 := compr(VB’ p- tpV(sample-s(2nd(RhoSig))) dot UB’, 1) /\
Kr2 := G(pair(M2, H’(T, Rho))) /\
enc-u(T,Rho,M2,2nd(Kr2)) == CUB /\ enc-v(T,Rho,M2,2nd(Kr2)) == CVB .

encaps-eve says that when eve has intercepted a message msg1 sent from A to B, eve fakes a
new message msg2, sends it to A, and computes a shared secret key with A. decaps-eve says
that when eve has faked a new message msg1, sent it to B, and B on his/her belief that the
message truly comes from A has replied to A a message msg2, eve intercepts the message msg2,
and computes a shared secret key with B.

5. Model checking and Man-In-The-Middle-Attack

We introduce the following search command:

search [1] in KYBER : init =>*
{(keys[alice]: key(K1,bob)) (keys[bob]: key(K2,alice))
(glean-keys: (key(K1,alice) key(K2,bob) KS)) OCs} .

where K1 and K2 are variables that denote arbitrary shared keys. K1 may or may not be equal to
K2. The command tries to find a state reachable from init such that: alice in her belief obtains
the shared key K1 with bob, bob in his belief obtains the shared key K2 with alice, and eve

owns both K1 and K2. Maude found a counterexample, and this kind of vulnerability belongs to
MITM attacks. Fig. 3 shows how this attack happens on Kyber, which is visualized from the
path leading to the counterexample Maude returned. There are mainly five steps as follows:

Step 1. Alice wants to construct a shared key with Bob, she starts by performing KEM.KeyGen(),
generating a public key 𝑝𝑘 and a secret key 𝑠𝑘. She keeps 𝑠𝑘, and sends 𝑝𝑘 to Bob.

Step 2. Eve intercepts the first message sent from Alice to Bob. She takes a random 𝑑𝑒, follows
the KEM.KeyGen() step to generate a pair (𝑝𝑘𝑒, 𝑠𝑘𝑒), and sends 𝑝𝑘𝑒 to Bob.

Step 3. Bob receives 𝑝𝑘𝑒 thinking it is from Alice. As a response, he takes a random𝑚0, performs
KEM.Enc(𝑝𝑘𝑒), and obtains a ciphertext 𝑐 and a shared key 𝐾𝑏. He sends the ciphertext 𝑐 back
to Alice, and keeps the key 𝐾𝑏, which he believes that it is the shared key obtained by him and
Alice.

Step 4. Eve intercepts the replied message which contains ciphertext 𝑐 sent from Bob to Alice.
She first performs KEM.Dec(𝑐, 𝑠𝑘𝑒) to obtain the shared key 𝐾𝑏. She then takes a random
𝑚𝑒0, performs KEM.Enc(𝑝𝑘), and obtains a ciphertext 𝑐𝑒 and a shared key 𝐾𝑎. She sends the
ciphertext 𝑐𝑒 back to Alice as a response for the first message.

Step 5. Alice receives the ciphertext 𝑐𝑒 thinking it is from Bob. She performs KEM.Dec(𝑐𝑒, 𝑠𝑘)
to obtain the shared key 𝐾𝑎. She believes that 𝐾𝑎 is the shared key obtained by her and Bob.

The reachable state space in the experiment is finite. Indeed, if we try to run the following
command: search in KYBER : init =>* {OCs} ., the number of returned solutions is finite,



Alice

𝑧 ← ℬ32
(𝑝𝑘, 𝑠𝑘′) = PKE.KeyGen()
𝑠𝑘 = (𝑠𝑘′||𝑝𝑘||H(𝑝𝑘)||𝑧)
return (𝑝𝑘, 𝑠𝑘)

pk−−→

𝑚′ = PKE.Dec(𝑐𝑒, s)
ce←−−

(𝐾̄
′
, 𝑟′) = G(𝑚′||H(𝑝𝑘))

𝑐′ = PKE.Enc(𝑝𝑘,𝑚′, 𝑟′)
if 𝑐𝑒 = 𝑐′ then
return𝐾𝑎 = KDF(𝐾̄

′
,H(𝑐𝑒))

Eve

𝑑𝑒 ← ℬ32
(𝜌𝑒, 𝜎𝑒) = G(𝑑𝑒)
𝑅𝑘×𝑘

𝑞 ∋ A𝑒 = generate(𝜌𝑒)

𝑅𝑘
𝑞 ∋ s𝑒, e𝑒 ← sampleCBD(𝜎𝑒)

t𝑒 = A𝑒s𝑒 + e𝑒
𝑝𝑘𝑒 = (t𝑒||𝜌𝑒)
𝑠𝑘𝑒 = s𝑒

return (𝑝𝑘𝑒, 𝑠𝑘𝑒)
pke−−−→

𝑚′
𝑏 = PKE.Dec(𝑐, s𝑒)

c←−−
(𝐾̄

′
𝑏, 𝑟

′
𝑏) = G(𝑚′

𝑏||H(𝑝𝑘𝑒))
𝑐′𝑏 = PKE.Enc(𝑝𝑘𝑒,𝑚′

𝑒, 𝑟
′
𝑒)

if 𝑐 = 𝑐′𝑏 then

return𝐾𝑏 = KDF(𝐾̄
′
𝑏,H(𝑐))

𝑚𝑒0 ← ℬ32
𝑚𝑒 = H(𝑚𝑒0)
(𝐾̄𝑒, 𝑟𝑒) = G(𝑚𝑒||H(𝑝𝑘))
𝑐𝑒 = PKE.Enc(𝑝𝑘,𝑚𝑒, 𝑟𝑒)
𝐾𝑎 = KDF(𝐾̄𝑒,H(𝑐𝑒))
return (𝑐𝑒,𝐾𝑎)

Bob

𝑚0 ← ℬ32
𝑚 = H(𝑚0)
(𝐾̄, 𝑟) = G(𝑚||H(𝑝𝑘𝑒))
𝑐 = PKE.Enc(𝑝𝑘𝑒,𝑚, 𝑟)
𝐾𝑏 = KDF(𝐾̄,H(𝑐))
return (𝑐,𝐾𝑏)

Figure 3: A counterexample found by Maude (note that we use PKE as an abbreviation for CPAPKE to
save space)

implying that the state space is finite. We give a brief explanation of this fact. Each state
is denoted as a braced associative-commutative soup of the ten observable components as
shown in Section 4. The key point is that the numbers of possible values that each observable
component (i.e., a name-value pair) can receive is finite. Indeed, 𝑝𝑠 in (prins : 𝑝𝑠) is always
(alice bob eve) because there is no rewrite rule that changes it. 𝑟𝑑𝑑𝑠 and 𝑟𝑑𝑚𝑠 in (rd-d :
𝑟𝑑𝑑𝑠) and (rd-m : 𝑟𝑑𝑚𝑠) never consist of more than (d1,d2) and (m1,m2), respectively, because
there is no rewrite rule that inserts element(s) into them. 𝑑0 and 𝑚0 in (d[p] : 𝑑0) and (m[p]
: 𝑚0) can only be in the sets {d1, d2} and {m1, m2}, respectively. Similarly, the numbers of
possible values for 𝑑𝑠 and 𝑚𝑠 in (ds : 𝑑𝑠) and (ms : 𝑚𝑠) are finite. 𝑚𝑠𝑔𝑠 in (nw : 𝑚𝑠𝑔𝑠) consists
of finite messages because (1) each of the two rewrite rules keygen and encaps adds a new
message into the network, but simultaneously it also removes one element from 𝑟𝑑𝑑𝑠 and 𝑟𝑑𝑚𝑠
(note that 𝑟𝑑𝑑𝑠 and 𝑟𝑑𝑚𝑠 never consist of more than (d1,d2) and (m1,m2) as shown above); (2)
the rewrite rule keygen-eve adds a new message msg2 into the network, but simultaneously it
also changes the status of an existing message msg1 from sent to intercepted, thus, keygen-eve



can only be applied finitely many times; and (3) the rewrite rule encaps-eve only adds a new
message msg2 into the network if that message does not exist before (note that the other rewrite
rules does not change the network or only update the status of messages). Similarly, 𝑘𝑒𝑦𝑠 in
(keys[p] : 𝑘𝑒𝑦𝑠) consists of finite entries because: the rewrite rule encaps removes one element
from 𝑟𝑑𝑚𝑠; and the rewrite rule decaps changes the status of an existing message msg2 from
sent to replied. In the same manner, we can show 𝑔𝑘𝑒𝑦𝑠 in (glean-keys : 𝑔𝑘𝑒𝑦𝑠) is finite. In
summary, we can conclude that the state space in our experiment is finite. Consequently, with
a search command to find a state satisfying some conditions, in finite time Maude will either
find no solutions or will find a state satisfying the conditions.

Remark. Readers may argue that this kind of attack is not a novel attack since Kyber is not
equipped with any feature for dealing with authentication. We agree on it. The paper instead
illustrates one symbolic approach for reasoning about KEMs rather than focusing on this kind
of attack. Our ultimate goal is to come up with a new security analysis/verification technique
for post-quantum cryptographic protocols, such as post-quantum TLS. Such protocols use
post-quantum cryptographic primitives, such as KEMs. Formally specifying such primitives is
necessary to analyze the security. What is described in the paper is our initial step toward the
goal.

Saber and SK-MLWR. In the same manner, we formalize the two other KEMs, i.e., Saber [4]
and SK-MLWR [5], specify them in Maude, and run Maude search command trying to find
the same kind of attack. The same MITM attacks are found by Maude. The reason is similar
to the Kyber case study, that is because there is no authentication, and thus, the intruder can
impersonate any party. We do not present these two case studies in this paper, but readers can
find the formal specifications on the webpage presented in Section 1.

6. Conclusion

We have presented an approach to security analysis of some lattice-based KEMs in the symbolic
model. We first used Maude as a specification language to formally specify the KEMs. After
that, by employing Maude search command, an MITM attack was found for each KEM. The
occurrence of the attack is basically because a KEM alone does not come with an authentication
solution.

Researchers have proposed a post-quantum TLS protocol [23] that uses a hybrid key exchange
method: a traditional key exchange algorithm together with a post-quantum KEM. The reason
why a post-quantum KEM is required is clear. However, why do we still need to employ a
traditional key exchange algorithm. One reason is that most post-quantum KEMs are not
studied/analyzed deeply, and thus, nothing guarantees that there is not any potential flaw
in them. Thus, deep security analysis of such KEMs in particular and other post-quantum
cryptographic primitives/protocols is an important challenge to guarantee their reliability. One
piece of our future work is to formally verify the security of the post-quantum TLS protocol
against both classical and quantum computers.

